Final exam / 2009.12.11 / Calculus for Applications / MAT 3243.001

Note Title 12/15/2009

1. In each case determine whether the limit exists, and if so, find the limit.

$$\text{(a)} \lim_{[x,y]\to 0} \frac{x^2y+y^3}{\sqrt{x^2+y^2}} \qquad \text{(b)} \lim_{[x,y]\to 0} \frac{x^3y+y^4}{x^4+y^4}$$

Check on Maple:

a) $\frac{(x^2+y^2)y}{\sqrt{x^2+y^2}} = \sqrt{x^2+y^2}\cdot y \to 0$

```
> (x^2*y+y^3)/sqrt(x^2+y^2); simplify(%);
```

$$\frac{x^2y+y^3}{\sqrt{x^2+y^2}}$$

$$y\sqrt{x^2+y^2}$$

b) if $x=0$ we get 1, if $y=0$, we get 0 $\therefore$ DNE

2. The temperature distribution (in degrees Fahrenheit) at position $[x, y]$ (in miles) is given by $T(x, y) = 98 - x^3y^2$. You start walking northwest from $[-1, 1]$ at 3 miles per hour. How fast is the temperature changing?

$$dT = -3x^2y^2dx - 2x^3y\,dy = \underbrace{[-3x^2y^2, -2x^3y]}_{\nabla T}\begin{bmatrix} dx \\ dy \end{bmatrix}$$

$$\nabla T[-1,1] = [-3, 2]$$

$$\hat{n} = \frac{1}{\sqrt{2}}\begin{bmatrix} -1 \\ 1 \end{bmatrix} \qquad D_{NW}T = \nabla T\cdot\hat{n} = \frac{1}{\sqrt{2}}[-3,2]\begin{bmatrix} -1 \\ 1 \end{bmatrix} = \frac{5}{\sqrt{2}}$$

$$\frac{dT}{dt} = D_{NW}T\cdot v = \frac{15}{\sqrt{2}} \approx 10.6\ {}^\circ F/\text{hour}$$

```
> with(linalg):
> T:=98-x^3*y^2;
```

$$T := 98 - x^3y^2$$

```
> v:=evalm(1/sqrt(2)*[-1,1]); evalf(%);
```

$$v := \left[-\frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2}\right]$$

$$[-0.7071067810, 0.7071067810]$$

```
> grad(T, [x,y]); subs({x=-1,y=1},%);
  sum(%[i]*v[i],i=1..2): %*3; evalf(%);
```

$$[-3x^2y^2, -2x^3y]$$

$$[-3, 2]$$

$$\frac{15\sqrt{2}}{2}$$

$$10.60660172$$

3. Let $f = \cos(1 + x^2 + y)$. Compute the Hessian matrix for f and find the quadratic Taylor approximation to f at the origin.

$$df = -\sin(1+x^2+y)\,[2x\,dx + dy] = \underbrace{-\sin(1+x^2+y)[2x, 1]}_{\nabla f} \begin{bmatrix} dx \\ dy \end{bmatrix}$$

$$\nabla f[0,0] = [0, -\sin(1)]$$

$$Hf = \begin{bmatrix} -\cos(1+x^2+y)\,4x^2 - \sin(1+x^2+y)\cdot 2 & -\cos(1+x^2+y)\,2x \\ -\cos(1+x^2+y)\,2x & -\cos(1+x^2+y) \end{bmatrix}$$

$$Hf[0,0] = \begin{bmatrix} -2\sin(1) & 0 \\ 0 & -\cos(1) \end{bmatrix} \qquad h = \begin{bmatrix} x \\ y \end{bmatrix}$$

$$f(x,y) \approx f(0,0) + Df[0,0] + \tfrac{1}{2} h^T Hf[0,0]\, h$$

$$= \cos(1) + [0, -\sin(1)]\begin{bmatrix} x \\ y \end{bmatrix} + \tfrac{1}{2}[x\ y]\begin{bmatrix} -2\sin(1) & 0 \\ 0 & -\cos(1) \end{bmatrix}\begin{bmatrix} x \\ y \end{bmatrix}$$

$$= \cos(1) - \sin(1)\,y - \sin(1)\,x^2 - \tfrac{1}{2}\cos(1)\,y^2$$

Alternately: $\cos(1+x^2+y) = \cos(1)\cos(x^2+y) - \sin(1)\sin(x^2+y)$

$$= \cos(1)\left(1 - \tfrac{1}{2}(x^2+y)^2 + \dots\right) - \sin(1)\left(x^2 + y - \dots\right)$$

$$= \cos(1) - \frac{\cos(1)\,y^2}{2} - \sin(1)\,x^2 - \sin(1)\,y + \text{higher order terms}$$

```
> cc:={x=0,y=0};
```

$cc := \{x = 0, y = 0\}$

```
> f:=cos(1+x^2+y); ff:=subs(cc,f);
```

$f := \cos(1 + x^2 + y)$

$ff := \cos(1)$

```
> grad(f, [x,y]); Df:=subs(cc,%);
```

$[-2\sin(1 + x^2 + y)\,x, -\sin(1 + x^2 + y)]$

$Df := [0, -\sin(1)]$

```
> hessian(f, [x,y]); Hf:=subs(cc,%);
```

$$\begin{bmatrix} -4\cos(1 + x^2 + y)\,x^2 - 2\sin(1 + x^2 + y) & -2\cos(1 + x^2 + y)\,x \\ -2\cos(1 + x^2 + y)\,x & -\cos(1 + x^2 + y) \end{bmatrix}$$

$$Hf := \begin{bmatrix} -2\sin(1) & 0 \\ 0 & -\cos(1) \end{bmatrix}$$

```
> h:=matrix([[x],[y]]);
```

$$h := \begin{bmatrix} x \\ y \end{bmatrix}$$

```
> ff+Df&*h+(1/2)*transpose(h)&*Hf&*h;
evalm(%); evalf(%);
```

$$\cos(1) + (Df \,\&^*\, h) + \left(\left(\left(\frac{1}{2}[x \quad y]\right) \&^*\, Hf\right) \&^*\, h\right)$$

$$\left[-\sin(1)\,y - x^2\sin(1) - \frac{1}{2}y^2\cos(1) + \cos(1)\right]$$

$[-0.8414709848\,y - 0.8414709848\,x^2 - 0.2701511530\,y^2 + 0.5403023059]$

4. A Petri dish 3 inches in diameter is used to grow a culture of H1N1 and the population density is given by $d(x, y) = x^2 + 2y^2 - y + 2$ in billions of virii per square inch. Where is the population density the lowest? The highest?

$\nabla d = [2x, 4y-1]$ $\quad \nabla d = 0 \Rightarrow x = 0,\ y = \frac{1}{4}$

Let $g = x^2 + y^2 - \frac{9}{4}$ $\quad \nabla d = \lambda \nabla g \Rightarrow 2x = \lambda 2x,\ 4y - 1 = \lambda 2y$

$\therefore x = 0$ or $\lambda = 1$

If $x = 0$, $y = \pm\frac{3}{2}$. If $\lambda = 1$, $4y - 1 = 2y$, $y = \frac{1}{2}$, $x = \pm\sqrt{2}$

	(x, y)	$d(x, y)$	
Crit	$[0, \frac{1}{4}]$	$15/8 = 1.875$	$\leftarrow$ min
Bdry Crit	$[0, \frac{3}{2}]$	5	
	$[0, -\frac{3}{2}]$	8	$\leftarrow$ max
	$[\sqrt{2}, \frac{1}{2}]$	4	
	$[-\sqrt{2}, \frac{1}{2}]$	4	

```
> f:=x^2+2*y^2-y+2;
```

$$f := x^2 + 2y^2 - y + 2$$

```
> Df:=grad(f, [x,y]);
```

$$Df := [2x, 4y - 1]$$

```
> solve(convert(Df,set),{x,y}):
  crit:=subs(%, [x,y]);
```

$$crit := \left[0, \frac{1}{4}\right]$$

```
> g:=x^2+y^2-(3/2)^2;
```

$$g := x^2 + y^2 - \frac{9}{4}$$

```
> L:=f-lambda*g;
```

$$L := x^2 + 2y^2 - y + 2 - \lambda\left(x^2 + y^2 - \frac{9}{4}\right)$$

```
> grad(L, [x,y,lambda]);
  solve(convert(%,set),{x,y,lambda}):
  seq(subs(%[i],[x,y]),i=1..3): seq(%[i],i=1..2),allvalues(%[3]);
  {crit} union {%}: convert(%,list);
  map(xx->subs({x=xx[1],y=xx[2]},f),%); evalf(%);
```

$$\left[2x - 2\lambda x, 4y - 1 - 2\lambda y, -x^2 - y^2 + \frac{9}{4}\right]$$

$$\left[0, \frac{3}{2}\right], \left[0, \frac{-3}{2}\right], \left[\sqrt{2}, \frac{1}{2}\right], \left[-\sqrt{2}, \frac{1}{2}\right]$$

$$\left[\left[0, \frac{1}{4}\right], \left[0, \frac{3}{2}\right], \left[0, \frac{-3}{2}\right], \left[\sqrt{2}, \frac{1}{2}\right], \left[-\sqrt{2}, \frac{1}{2}\right]\right]$$

$$\left[\frac{15}{8}, 5, 8, 4, 4\right]$$

$$[1.875000000, 5., 8., 4., 4.]$$

5. Sketch the solid enclosed by the surfaces $z = 3 - \sqrt{x^2 + y^2}$ and $z = 0$. Use triple integration in cylindrical coordinates to compute its volume.



$$\int_0^3\int_{-\pi}^{\pi}\int_0^{3-r} r\,dz\,d\theta\,dr = 2\pi\int_0^3 rz\Big|_0^{3-r}\,dr$$

$$= 2\pi\int_0^3 \underbrace{(3-r)\,r}_{3r-r^2}\,dr = 2\pi\left[\frac{3}{2}r^2-\frac{r^3}{3}\right]_0^3 = 2\pi\left[\frac{3^3}{2}-3^2\right]$$

$$= 2\pi\,3^2\left[\frac{3}{2}-1\right] = \boxed{9\pi}$$

```
> X:=[r*cos(theta),r*sin(theta),z];
```

$$X := [r\cos(\theta), r\sin(\theta), z]$$

```
> z=3-r; solve(%,r);
  plot3d(%,theta=-Pi..Pi,z=0..3,coords=cylindrical,axes=boxed,style=
  patchcontour,scaling=constrained);
```

$$z = 3 - r$$
$$-z + 3$$




```
> jacobian(X,[r,theta,z]); det(%): simplify(%);
  int(int(int(%,z=0..3-r),theta=-Pi..Pi),r=0..3);
```

$$\begin{bmatrix} \cos(\theta) & -r\sin(\theta) & 0 \\ \sin(\theta) & r\cos(\theta) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$r$$
$$9\pi$$

6. Find the scalar potential for the vector field $F = [z\cos x, 2y, \sin x]$ or show that such a potential doesn't exist.

$$F_x = z\cos x \Rightarrow F = z\sin x + g(y,z)$$

$$F_y = 2y = g_y \quad \therefore\ g(y,z) = y^2 + h(z)$$

$$\therefore\ F = z\sin x + y^2 + h(z)$$

$$F_z = \sin x = \sin x + h'(z) \quad \therefore\ h'(z) = 0 \quad \therefore\ h(z) = \text{const.}$$

$$\therefore\ \boxed{F = z\sin x + y^2 + C}$$

```
> z*sin(x)+y^2; grad(%,[x,y,z]);
```

$$z\sin(x) + y^2$$

$$[z\cos(x), 2y, \sin(x)]$$

7. Integrate $\omega = 2x\,dy - 3y\,dx$ around the circle of radius 3 centered at the origin counterclockwise. Compute the same integral using Green's theorem.

$$\begin{bmatrix} x \\ y \end{bmatrix} = 3\begin{bmatrix} \cos\theta \\ \sin\theta \end{bmatrix} \qquad \begin{bmatrix} dx \\ dy \end{bmatrix} = 3\begin{bmatrix} -\sin\theta \\ \cos\theta \end{bmatrix} d\theta$$

$$\omega = 2\cdot 3\cos\theta\cdot 3\cos\theta\, d\theta - 3\cdot 3\sin\theta\cdot(-3\sin\theta)\, d\theta$$

$$= 9\left[2(\cos\theta)^2 + 3(\sin\theta)^2\right] d\theta$$

$$\oint \omega = 9\int_{-\pi}^{\pi}\Big(2\underbrace{(\cos\theta)^2}_{} + 3\underbrace{(\sin\theta)^2}_{}\Big)\,d\theta = 9\left[2\cdot\tfrac{1}{2} + 3\cdot\tfrac{1}{2}\right]2\pi = \boxed{45\pi}$$

on average $\frac{1}{2}$

$$d\omega = 2\,dx\,dy - 3\,dy\,dx = 5\,dx\,dy$$

$$\iint d\omega = 5\iint dx\,dy = 5\cdot\text{Area} = 5\pi 3^2 = \boxed{45\pi}$$

```
> w:=2*x*dy-3*y*dx;
```

$$w := 2x\,dy - 3y\,dx$$

```
> X:=[3*cos(theta),3*sin(theta)];
```

$$X := [3\cos(\theta), 3\sin(\theta)]$$

```
> DX:=diff(X,theta);
```

$$DX := [-3\sin(\theta), 3\cos(\theta)]$$

```
> subs({x=X[1],y=X[2],dx=DX[1],dy=DX[2]},w);
  int(%,theta=-Pi..Pi);
```

$$18\cos(\theta)^2 + 27\sin(\theta)^2$$

$$45\pi$$

```
> curl([-3*y,2*x,0],[x,y,z]);
  %[3]*r; int(int(%,r=0..3),theta=-Pi..Pi);
```

$$[0, 0, 5]$$

$$5r$$

$$45\pi$$

8. Compute the flux of $F = [x, y, z]$ through the surface $z = 1 - x^2 - y^2, z \geq 0$ oriented with the upward normal both directly and also using the divergence theorem.

$$\begin{bmatrix} x \\ y \\ z \end{bmatrix} = \begin{bmatrix} r\cos\theta \\ r\sin\theta \\ 1-r^2 \end{bmatrix} \qquad \begin{bmatrix} dx \\ dy \\ dz \end{bmatrix} = \begin{bmatrix} \cos\theta\, dr - r\sin\theta\, d\theta \\ \sin\theta\, dr + r\cos\theta\, d\theta \\ -2r\, dr \end{bmatrix}$$



$$d\bar{S} = \begin{bmatrix} dy\,dz \\ dz\,dx \\ dx\,dy \end{bmatrix} = \underbrace{\begin{bmatrix} 2r^2\cos\theta \\ 2r^2\sin\theta \\ r \end{bmatrix}}_{N} dr\,d\theta$$

$$\int \bar{F}\cdot d\bar{S} = \int_{-\pi}^{\pi}\int_0^1 [r\cos\theta, r\sin\theta, 1-r^2] \begin{bmatrix} 2r^2\cos\theta \\ 2r^2\sin\theta \\ r \end{bmatrix} dr\,d\theta$$

$$= \int_{-\pi}^{\pi}\int_0^1 \Big(\underbrace{2r^3(\cos\theta)^2 + 2r^3(\sin\theta)^2}_{2r^3} + r - r^3\Big)\,dr\,d\theta$$

$$= 2\pi\left[\frac{r^4}{4} + \frac{r^2}{2}\right]_0^1 = 2\pi\left[\frac{1}{4}+\frac{1}{2}\right] \boxed{= \frac{3}{2}\pi}$$

The same integral but over the unit disc in the x-y plane is 0, because $\hat{n} \parallel \hat{k}$, while the z component of F is 0 on the disc.

By the divergence theorem $\iiint \operatorname{div} F\, dV = \underbrace{\iint_{\text{paraboloid}} \bar{F}\cdot d\bar{S}}_{\frac{3}{2}\pi} + \underbrace{\iint_{\text{disc}} \bar{F}\cdot d\bar{S}}_{0}$

$\operatorname{div} F = 3$, so $\iiint \operatorname{div} F \cdot dV$

$= 3 \cdot \text{Volume(solid paraboloid)}$

$$= 3\int_0^1\int_{-\pi}^{\pi}\int_0^{1-r^2} r\,dz\,d\theta\,dr = 3\cdot 2\pi\int_0^1 rz\Big|_0^{1-r^2} dr = 6\pi\int_0^1 (r - r^3)\,dr$$

$$= 6\pi\left(\frac{r^2}{2} - \frac{r^4}{4}\right)\Big|_0^1 = 6\pi\left(\frac{1}{2}-\frac{1}{4}\right) \boxed{= \frac{3\pi}{2}}$$

```
> F:=[x,y,z];
```

$$F := [x, y, z]$$

```
> X:=[r*cos(theta),r*sin(theta),1-r^2];
```

$$X := [r\cos(\theta), r\sin(\theta), 1 - r^2]$$

```
> jacobian(X,[r,theta]);
  crossprod(col(%,1),col(%,2)): N:=simplify(%);
```

$$\begin{bmatrix} \cos(\theta) & -r\sin(\theta) \\ \sin(\theta) & r\cos(\theta) \\ -2r & 0 \end{bmatrix}$$

$$N := [2r^2\cos(\theta), 2r^2\sin(\theta), r]$$

```
> subs({x=X[1],y=X[2],z=X[3]},F);
  sum(%[i]*N[i],i=1..3): simplify(%): expand(%);
  int(int(%,r=0..1),theta=-Pi..Pi);
```

$$[r\cos(\theta), r\sin(\theta), 1 - r^2]$$

$$r^3 + r$$

$$\frac{3\pi}{2}$$

Same but for the unit disc in the x-y plane

```
> X:=[r*cos(theta),r*sin(theta),0];
```

$$X := [r\cos(\theta), r\sin(\theta), 0]$$

```
> jacobian(X,[r,theta]);
  crossprod(col(%,1),col(%,2)): N:=simplify(%);
```

$$\begin{bmatrix} \cos(\theta) & -r\sin(\theta) \\ \sin(\theta) & r\cos(\theta) \\ 0 & 0 \end{bmatrix}$$

$$N := [0, 0, r]$$

```
> subs({x=X[1],y=X[2],z=X[3]},F);
  sum(%[i]*N[i],i=1..3): simplify(%): expand(%);
  int(int(%,r=0..1),theta=-Pi..Pi);
```

$$[r\cos(\theta), r\sin(\theta), 0]$$

$$0$$

$$0$$

Now using the divergence theorem:

```
> diverge(F,[x,y,z]);
  int(int(int(%*r,z=0..1-r^2),theta=-Pi..Pi),r=0..1);
```

$$3$$

$$\frac{3\pi}{2}$$